Topologies and Closures are Equivalent

Greg O'Keefe
Automated Reasoning Group, RSISE,
Australian National University
greg.okeefe@anu.edu.au
http://axiom.anu.edu.au/∼okeefe

December 12, 2003

This Isabelle/HOL/Isar [2] theory formalises parts of [1, Chapter 3]. The aim is to show that topological spaces and closure spaces are "the same thing". That is, the open sets of a closure space are a topology, and the topological closure operator of a topological space is a closure. Furthermore, these "conversions" or functors, are inverse.

I have attempted to follow [1] as closely as possible. In future work, I hope to investigate the differences and consider whether they constitute problems with Isar/HOL, problems with the informal mathematics, or fundamental incompatabilities between the two approaches.

Contents

theory *closure* = *FuncSet*:

1 Topological Spaces

This locale implements [1, Definitions 2.1, 2.3, 2.5 and 4.2].

locale *topological-space* = *var X* + *var T* +
— axioms from Definition 2.1

assumes *non-empty*:
$X \neq \{\}$
and *carrier-subset*:
$\forall A \in T.\ A \subseteq X$
and *universe-in*:
$X \in T$
and *empty-in*:
$\{\} \in T$
and *finite-intersect*:
$\forall F.\ F \subseteq T\ \&\ \mathit{finite}\ F\ \&\ F \neq \{\} \longrightarrow \bigcap F \in T$
and *arbitrary-union*:
$\forall F.\ F \subseteq T \longrightarrow \bigcup F \in T$

— Definition 2.3
fixes *neighbourhood-of* :: $'a\ set \Rightarrow\ 'a \Rightarrow bool$ (- *neighbourhood of* - [*60*,*60*] *80*)
defines N *neighbourhood of* $x \equiv x \in X\ \&\ (\exists A \in T.\ x \in A\ \&\ A \subseteq N)$

— Definition 2.5
fixes *t-closed* :: $'a\ set \Rightarrow bool$
defines *t-closed* $A \equiv \exists A' \in T.\ A = X - A'$

— Definition 4.2
fixes *topo-closure* :: $'a\ set \Rightarrow\ 'a\ set$ (-- [*90*] *80*)
defines *topo-closure* $A \equiv \{x.\ x \in X\ \&\ (\forall N.\ N$ *neighbourhood of* $x \longrightarrow N \cap A \neq \{\})\}$

Here are a few results that are too obvious to mention in an informal text.

lemma (**in** *topological-space*) *in-its-neighbourhood*:
assumes N *neighbourhood of* x
shows $x \in N$
by (*simp*!) *auto*

lemma (**in** *topological-space*) *has-a-neighbourhood*:
$\forall x \in X.\ \exists N.\ N$ *neighbourhood of* x
proof−
from *universe-in* **show** *?thesis*
by (*unfold neighbourhood-of-def*) *auto*
qed

lemma (**in** *topological-space*) *t-closure-subset*:
assumes $A \subseteq X$
shows $\overline{A} \subseteq X$
by (*unfold topo-closure-def*) *auto*

lemma (**in** *topological-space*) *t-subset-closure*:
assumes *A-subset*: $A \subseteq X$

shows $A \subseteq \overline{A}$
proof (*rule*, *simp only*: *topo-closure-def mem-Collect-eq*, *intro conjI allI impI*)
fix x **and** N
assume $x{\in}A$
with *A-subset* **show** $x{\in}X$
by *auto*
fix N
assume N *neighbourhood of* x
with *in-its-neighbourhood* **have** $x \in N{\cap}A$
by *auto*
thus $N{\cap}A \neq \{\}$
by *auto*
qed

lemma (**in** *topological-space*) *t-closed-subset* [*simp*]:
assumes *A-closed*: *t-closed A*
shows $A \subseteq X$
proof−
from *A-closed* **obtain** A' **where** $A'{\in}T$ **and** $A{=}X{-}A'$
by (*unfold t-closed-def*) *auto*
thus *?thesis* **by** *auto*
qed

lemma (**in** *topological-space*) *comp-closed-open*:
assumes *A-closed*: *t-closed A*
shows $X{-}A \in T$
proof−
from *A-closed* **obtain** A' **where**
A'-in: $A'{\in}T$ **and**
A-is: $X{-}A'{=}A$
by (*unfold t-closed-def*) *auto*
from *A'-in* **and** *A-is* **and** *carrier-subset*
have $A'{=}X{-}A$
by *auto*
with *A'-in* **show** *?thesis*
by *auto*
qed

lemma(**in** *topological-space*) *pairwise-union*:
assumes *a*: *t-closed A*
and *b*: *t-closed B*
shows *t-closed* $(A{\cup}B)$
proof−
from *a* **and** *b* **obtain** A' **and** B'
where $A' \in T$ **and** *a'*: $A = X{-}A'$

and $B' \in T$ **and** b': $B = X - B'$
by (*unfold t-closed-def*) *auto*
hence $\{A',B'\} \subseteq T \wedge$ *finite* $\{A',B'\} \wedge \{A',B'\} \neq \{\}$
by *auto*
with *finite-intersect* **have** $\bigcap\{A',B'\} \in T$
by (*elim allE impE*)
hence *1*: $A' \cap B' \in T$
by *auto*
from a' **and** b' **have** *2*: $A \cup B = X - (A' \cap B')$
by *auto*
from *1* **and** *2* **show** *t-closed* $(A \cup B)$
by (*unfold t-closed-def*) *auto*
qed

These results are mostly from Mendelsons text, with the remainder being lemmas required to show these. The first is an exercise.

lemma (**in** *topological-space*) *ex-two-four*:
assumes *Fs-closed*: $\forall A \in F.$ *t-closed* A
and *F-nonempty*: $F \neq \{\}$
shows *t-closed* $(\bigcap F)$
proof−
have *F-subsets*: $\forall A \in F.\ A \subseteq X$
by (*simp*!)
with *F-nonempty* **have** *UF-is*: $\bigcap F = X - \bigcup\{X - B \mid B.\ B \in F\}$
by *auto*
from *Fs-closed* **and** *comp-closed-open*
have $\forall D \in \{X - B \mid B.\ B \in F\}.\ D \in T$ **by** *auto*
hence $\{X - B \mid B.\ B \in F\} \subseteq T$
by *auto*
with *arbitrary-union* **have** $\bigcup\{X - A \mid A.\ A \in F\} \in T$
by *auto*
with *UF-is* **show** *?thesis*
by (*unfold t-closed-def*) *auto*
qed

lemma (**in** *topological-space*) *four-three*:
assumes $A \subseteq X$
and *t-closed* F
and $A \subseteq F$
shows $\overline{A} \subseteq F$
proof(*rule*, *erule contrapos-pp*)
fix x
assume *xnF*:$x \notin F$
show $x \notin \overline{A}$
proof (*cases* $x \in X$)
assume *xX*: $x \in X$
with *xnF* **have** *xXF*: $x \in X - F$ **by** *auto*

```
    from comp-closed-open have XF-open: X−F ∈ T
      by (simp!)
    have A⊆F .
    hence XF-XA: X−F ⊆ X−A by auto
    hence empty-inter: (X−F) ∩ A = {} by auto
    from xX and xXF and XF-open and XF-XA
    have (X−F) neighbourhood of x
      by (simp!) auto
    with empty-inter
    show x ∉ A̅
      by (unfold neighbourhood-of-def topo-closure-def) auto
  next
    assume x∉X
    thus x ∉ A̅
      by (unfold topo-closure-def) auto
  qed
qed


lemma (in topological-space) four-four:
  assumes A-subset: A⊆X
  and x ∈ X
  and x ∉ A̅
  shows ∃F. t-closed F & A⊆F & x∉F
proof−
  have x ∉ A̅ and x ∈X .
    from this obtain N and B
    where 1:N neighbourhood of x
      and 2:N ∩ A = {}
      and 3:x∈B
      and 4:B∈T
      and 5:B⊆N
      by (unfold neighbourhood-of-def topo-closure-def) auto
    hence 6:B ∩ A = {}
      by auto
    from 4 have 7:t-closed (X−B)
      by (unfold t-closed-def) auto
    from A-subset and 6
    have 8:A ⊆ (X−B)
      by auto
    with 3
    have 9:x∉(X−B)
      by auto
    from 7 and 8 and 9
    show ?thesis
      by auto
qed
```

theorem (**in** *topological-space*) *four-five*:
  assumes *subset*: $A \subseteq X$
  shows $\overline{A} = \bigcap\{B.\ t\text{-}closed\ B\ \&\ A \subseteq B\}$
proof
  show $\overline{A} \subseteq \bigcap\{B.\ t\text{-}closed\ B \wedge A \subseteq B\}$
  proof (*rule Inter-greatest*)
    fix B
    assume $B \in \{B.\ t\text{-}closed\ B \wedge A \subseteq B\}$
    hence *t-closed* B **and** $A \subseteq B$ **by** *simp-all*
    with *subset* **show** $\overline{A} \subseteq B$
      by (*rule four-three*)
  qed
next
  show $\bigcap\{B.\ t\text{-}closed\ B \wedge A \subseteq B\} \subseteq \overline{A}$
  proof
    fix x
    assume *lhs*: $x \in \bigcap\{B.\ t\text{-}closed\ B \wedge A \subseteq B\}$
    from *empty-in* **have** *t-closed* X
      by (*unfold t-closed-def*) *auto*
    with *lhs* **and** *subset* **have** *x-in*: $x \in X$
      by *auto*
    show $x \in \overline{A}$
    proof (*rule ccontr*)
      assume $x \notin \overline{A}$
      with *subset* **and** *x-in* **have** $\exists F.\ t\text{-}closed\ F\ \&\ A \subseteq F\ \&\ x \notin F$
        by (*rule four-four*)
      with *lhs* **show** *False*
        by *auto*
    qed
  qed
qed

The following is established in the text between Theorems 4.5 and 4.6: $\overline{A}$ "being the intersection of closed sets, is itself a closed set"

lemma (**in** *topological-space*) *closure-closed*:
  assumes *A-subset*: $A \subseteq X$
  shows *t-closed* $(\overline{A})$
proof–
  have *2*: $\forall B \in \{B.\ t\text{-}closed\ B\ \&\ A \subseteq B\}.\ t\text{-}closed\ B$
    by *auto*
  from *empty-in* **and** *A-subset*
  have $X \in \{B.\ t\text{-}closed\ B\ \&\ A \subseteq B\}$
    by (*unfold t-closed-def*) *auto*
  hence $\{B.\ t\text{-}closed\ B\ \&\ A \subseteq B\} \neq \{\}$
    by *auto*
  from *2* **and** *this* **have** *t-closed* $(\bigcap\{B.\ t\text{-}closed\ B\ \&\ A \subseteq B\})$
    by (*rule ex-two-four*)
  with *A-subset* **and** *four-five* **show** *?thesis*

```
    by auto
  qed

theorem (in topological-space) four-six:
  assumes A-subset: A ⊆ X
  shows (t-closed A) = (Ā = A)
proof
  assume 1: Ā = A
  from A-subset have t-closed (Ā)
    by (rule closure-closed)
  with 1 show t-closed A
    by simp
next
  assume A-closed: t-closed A
  show Ā = A
  proof (rule equalityI)
    from A-closed and A-subset
    have A ∈ {B. t-closed B & A ⊆ B}
      by auto
    with A-subset and four-five show Ā ⊆ A
      by auto
  next
    from A-subset show A ⊆ Ā
      by (rule t-subset-closure)
  qed
qed

lemma (in topological-space) four-seven-CL4:
  assumes A ⊆ X
  and B ⊆ X
  shows -(A ∪ B) = Ā ∪ -B
proof (rule sym, rule equalityI, simp only: Un-subset-iff, intro conjI subsetI)
  fix x
  assume xCA: x ∈ Ā
  hence xX: x ∈ X
    by (unfold topo-closure-def) auto
  from xCA have ∀ N. N neighbourhood of x ⟶ N∩A ≠ {}
    by (unfold topo-closure-def) auto
  hence ∀ N. N neighbourhood of x ⟶ N∩(A∪B) ≠ {}
    by auto
  with xX show x ∈ -(A∪B)
    by (unfold topo-closure-def) auto
next
  fix x
  assume xCB: x ∈ -B
  hence xX: x ∈ X
```

```
    by (unfold topo-closure-def) auto
  from xCB have ∀ N. N neighbourhood of x ⟶ N∩B ≠ {}
    by (unfold topo-closure-def) auto
  hence ∀ N. N neighbourhood of x ⟶ N∩(A∪B) ≠ {}
    by auto
  with xX
  show x ∈ -(A∪B)
    by (unfold topo-closure-def) auto
next
  have A ⊆ A̅
    by (rule t-subset-closure)
  moreover have B ⊆ -B
    by (rule t-subset-closure)
  ultimately have subset: A ∪ B ⊆ A̅ ∪ -B
    by auto
  have t-closed (A̅)
    by (rule closure-closed)
  moreover have t-closed (-B)
    by (rule closure-closed)
  moreover note pairwise-union
  ultimately have closed: t-closed (A̅ ∪ -B)
    by auto
  have A⊆X and B⊆X .
  hence A ∪ B ⊆ X
    by auto
  with subset and closed and four-three
  show ̅(A∪B) ⊆ A̅ ∪ -B
    by auto
qed
```

2 Closure Spaces

[1, Definitions 4.8, 4.9, Corollary 4.12 (defn)]. Mendelson does not specify that the carrier should be non-empty, though he does for topological spaces.

```
locale closure-space = var X + var Cl +
  — Definition 4.8, which refers back to theorem 4.7
  assumes non-empty:
    X ≠ {}
  and empty-fixed:
    Cl {} = {}
  and carrier-fixed:
    Cl X = X
  and non-decreasing:
    ∀ A. A⊆X ⟶ A ⊆ Cl A
  and dist-union [simp]:
    ∀ A B. A⊆X & B⊆X ⟶ Cl (A ∪ B) = Cl A ∪ Cl B
  and idempotent:
```

$\forall A.\ A \subseteq X \longrightarrow Cl\ (Cl\ A) = Cl\ A$

— Definition 4.9
fixes *c-closed* :: *'a set* $\Rightarrow$ *bool*
defines *c-closed* $A \equiv A \subseteq X\ \&\ Cl\ A = A$

— this definition is packed into Corollary 4.12
fixes *c-open* :: *'a set* $\Rightarrow$ *bool*
defines *c-open* $A \equiv A \subseteq X\ \&$ *c-closed* $(X - A)$

theorem (**in** *topological-space*) *four-seven*:
closure-space X topo-closure
proof (*intro closure-space.intro*)
from *non-empty* **show** $X \neq \{\}$ **.**
show $\forall A.\ A \subseteq X \longrightarrow A \subseteq \overline{A}$
by (*intro allI impI*, *rule t-subset-closure*)
have $-X \subseteq X$ **by** (*rule t-closure-subset*, *auto*)
moreover have $X \subseteq -X$ **by** (*rule t-subset-closure*, *auto*)
ultimately show $-X = X$ **..**
show $-\{\} = \{\}$
proof−
have $N \cap \{\} = \{\}$ **by** *auto*
with *has-a-neighbourhood*
have $\forall x \in X.\ x \notin -\{\}$
by (*unfold topo-closure-def*) *auto*
with *t-closure-subset*
have $\forall x.\ x \notin -\{\}$
by *auto*
thus *?thesis*
by *auto*
qed
show $\forall A.\ A \subseteq X \longrightarrow -(\overline{A}) = \overline{A}$
proof (*intro allI impI*)
fix A
assume *A-subset*: $A \subseteq X$
hence $\overline{A} \subseteq X$
and *t-closed* $(\overline{A})$
by (*rule t-closure-subset*) (*rule closure-closed*)
thus $-(\overline{A}) = \overline{A}$
by (*simp only*: *four-six*)
qed
from *four-seven-CL4*
show $\forall A\ B.\ A \subseteq X \wedge B \subseteq X \longrightarrow -(A \cup B) = \overline{A} \cup -B$
by *auto*
qed

Note that Mendelson does not specify that A and B are subsets of the carrier

X, but it is needed. Otherwise, this proof is quite close to Mendelsons.

lemma (**in** *closure-space*) *four-ten*:
  assumes *a*: $A \subseteq X$ **and** *b*: $B \subseteq X$
  and *ab*: $A \subseteq B$
  shows $Cl\ A \subseteq Cl\ B$
proof−
  from *ab* **have** $A \cup B = B$ **by** *auto*
  hence $Cl\ B = Cl\ (A \cup B)$ **by** *auto*
  also from *a* **and** *b* **have** $\ldots = (Cl\ A) \cup (Cl\ B)$
    by *simp*
  finally have $(Cl\ A) \cup (Cl\ B) = Cl\ B$ **by** (*rule sym*)
  thus $Cl\ A \subseteq Cl\ B$ **by** *auto*
qed

lemma (**in** *closure-space*) *c-closure-subset*:
  assumes $A \subseteq X$
  shows $Cl\ A \subseteq X$
proof−
  have $Cl\ A \subseteq Cl\ X$
    by (*rule four-ten*) *simp-all*
  with *carrier-fixed*
  show $Cl\ A \subseteq X$
    by *auto*
qed

lemma (**in** *closure-space*) *finite-union-c-closed*:
  assumes *f*: *finite F*
  and *ss*: $F \subseteq \{A.\ c\text{-}closed\ A\}$
  shows *c-closed* $(\bigcup F)$
proof−
  from *f* **and** *ss* **show** *?thesis*
  proof *induct*
    from *empty-fixed*
    show *c-closed* $(\bigcup \{\})$
      by (*simp add*: *c-closed-def*)
  next
    fix $F'\ B$
    assume *insert* $B\ F' \subseteq \{A.\ c\text{-}closed\ A\}$
    hence *B-closed*: *c-closed B*
      and *F′-ss*: $F' \subseteq \{A.\ c\text{-}closed\ A\}$
      by *auto*
    assume $F' \subseteq \{A.\ c\text{-}closed\ A\} \Longrightarrow c\text{-}closed\ (\bigcup F')$
    with *F′-ss* **have** *UF′-closed*: *c-closed* $(\bigcup F')$
      by *auto*
    with *B-closed* **have** *B-css*: $B \subseteq X$ **and** *UF′-css*: $(\bigcup F') \subseteq X$

by (*simp-all add*: *c-closed-def*)
with *dist-union* **have** $Cl\ (B \cup (\bigcup F')) = Cl\ B \cup Cl\ (\bigcup F')$
by *auto*
also from *B-closed* **and** *UF′-closed*
have $\ldots = B \cup (\bigcup F')$
by (*unfold c-closed-def*) *auto*
finally have $Cl\ (B \cup (\bigcup F')) = B \cup \bigcup F'$ **.**
with *B-css* **and** *UF′-css* **have** *c-closed* $(B \cup (\bigcup F'))$
by (*simp add*: *c-closed-def*) *auto*
thus *c-closed* ($\bigcup$ *insert B F′*)
by *auto*
qed
qed

lemma (**in** *closure-space*) *intersection-c-closed*:
assumes *subset*: $F \subseteq \{A.\ c\text{-}closed\ A\}$
and *non-empty*: $F \neq \{\}$
shows *c-closed* $(\bigcap F)$
proof−
from *subset*
have *all-fixed*: $\forall A \in F.\ Cl\ A = A$
by (*unfold c-closed-def*) *auto*
from *subset* **have** *all-subsets*: $\forall A \in F.\ A \subseteq X$
by (*unfold c-closed-def*) *auto*
with *non-empty* **have** *inter-subset*: $\bigcap F \subseteq X$
by *auto*
have $\forall A \in F.\ \bigcap F \subseteq A$
by *auto*
with *four-ten* **and** *all-subsets* **and** *inter-subset*
have $\forall A \in F.\ Cl\ (\bigcap F) \subseteq Cl\ A$
by *auto*
with *all-fixed* **have** *l2r*: $Cl\ (\bigcap F) \subseteq \bigcap F$
by *auto*
from *non-decreasing* **and** *inter-subset*
have *r2l*: $\bigcap F \subseteq Cl\ (\bigcap F)$
by *auto*
from *l2r* **and** *r2l* **have** $\bigcap F = Cl\ (\bigcap F)$
by *auto*
with *inter-subset* **show** *c-closed* $(\bigcap F)$
by (*unfold c-closed-def*) *auto*
qed

theorem (**in** *closure-space*) *four-eleven*:
shows *c-closed* {}
and *c-closed X*
and $\forall F.\ (\forall A \in F.\ c\text{-}closed\ A)\ \&\ finite\ F \longrightarrow c\text{-}closed\ (\bigcup F)$
and $\forall F.\ (\forall A \in F.\ c\text{-}closed\ A)\ \&\ F \neq \{\} \longrightarrow c\text{-}closed\ (\bigcap F)$

proof−
  from *empty-fixed* **show** *empty-closed*: *c-closed* {}
    by (*simp add*: *c-closed-def*)
  from *carrier-fixed* **show** *c-closed X*
    by (*simp add*: *c-closed-def*)
  from *finite-union-c-closed*
  show $\forall F.\ (\forall A{\in}F.\ \textit{c-closed}\ A) \wedge \textit{finite}\ F \longrightarrow \textit{c-closed}\ (\bigcup F)$
    by *auto*
  from *intersection-c-closed*
  show $\forall F.\ (\forall A{\in}F.\ \textit{c-closed}\ A)\ \&\ F{\neq}\{\} \longrightarrow \textit{c-closed}\ (\bigcap F)$
    by *auto*
qed

lemma (**in** *closure-space*) *finite-inter-open*:
  assumes *Fs-open*: $F \subseteq \{A.\ \textit{c-open}\ A\}$
  and *F-finite*: *finite F*
  and *non-empty*: $F{\neq}\{\}$
  shows *c-open* $(\bigcap F)$
proof−
  from *Fs-open* **have** *closed*: $\forall B{\in}\{X{-}A \mid A.\ A{\in}F\}.\ \textit{c-closed}\ B$
    by (*unfold c-open-def*) *auto*
  from *F-finite* **have** *finite* $((\lambda A.\ X{-}A)\ \text{‘}\ F)$
    by (*rule finite-imageI*)
  moreover have $((\lambda A.\ X{-}A)\ \text{‘}\ F) = \{X{-}A \mid A.\ A{\in}F\}$
    by *auto*
  ultimately have *finite*: *finite* $\{X{-}A \mid A.\ A{\in}F\}$
    by *simp*
  have $\bigcup\{X{-}A \mid A.\ A{\in}F\} \subseteq X$
    by *auto*
  hence *comp-comp*: $X{-}(X{-}\bigcup\{X{-}A \mid A.\ A{\in}F\}) = \bigcup\{X{-}A \mid A.\ A{\in}F\}$
    by *auto*
  have $\forall F.\ (\forall A{\in}F.\ \textit{c-closed}\ A)\ \&\ \textit{finite}\ F \longrightarrow \textit{c-closed}\ (\bigcup F)$
    by (*rule four-eleven*)
  with *closed* **and** *finite*
  have *c-closed* $(\bigcup\ \{X{-}A \mid A.\ A{\in}F\})$
    by *auto*
  with *comp-comp* **have** *open-thing*: *c-open* $(X{-}\bigcup\ \{X{-}A \mid A.\ A{\in}F\})$
    by (*unfold c-open-def*) *auto*
  from *Fs-open* **and** *non-empty*
  have $X - \bigcup\{X{-}A \mid A.\ A{\in}F\} = \bigcap F$
    by (*unfold c-open-def*) *auto*
  with *open-thing* **show** *c-open* $(\bigcap F)$
    by *simp*
qed

lemma (**in** *closure-space*) *union-open*:

assumes *Fs-open*: $F \subseteq \{A.\ c\text{-}open\ A\}$
shows *c-open* $(\bigcup F)$
proof (*cases* $F=\{\}$)
assume $F=\{\}$
hence *empty*: $\bigcup F=\{\}$
by *auto*
have *c-closed* X
by (*rule four-eleven*)
hence *c-open* $\{\}$
by (*unfold c-open-def*) *auto*
thus *c-open* $(\bigcup F)$
by (*simp add*: *empty*)
next
— complements are closed
from *Fs-open* **have** *closed*: $\forall B \in \{X-A \mid A.\ A\in F\}.$ *c-closed* B
by (*unfold c-open-def*) *auto*
— intersection closed
moreover have $\forall F.\ (\forall A\in F.$ *c-closed* $A)$ & $F\neq\{\} \longrightarrow$ *c-closed* $(\bigcap F)$
by (*rule four-eleven*)
ultimately have
closed-thing: $F\neq\{\} \longrightarrow$ *c-closed* $(\bigcap \{X-A \mid A.\ A\in F\})$
by *auto*
— complement open
assume *non-empty*: $F \neq \{\}$
hence $\bigcap \{X-A \mid A.\ A\in F\} \subseteq X$
by *auto*
hence *comp-comp*: $X-(X-\bigcap\{X-A \mid A.\ A\in F\}) = \bigcap\{X-A \mid A.\ A\in F\}$
by *auto*
have *comp-subset*: $X-\bigcap\{X-A \mid A.\ A\in F\} \subseteq X$
by *auto*
from *non-empty* **and** *closed-thing* **and** *comp-comp*
have *closed-comp-comp*: *c-closed* $(X-(X-\bigcap\{X-A \mid A.\ A\in F\}))$
by *auto*
from *comp-subset* **and** *closed-comp-comp*
have *open-thing*: *c-open* $(X - \bigcap\{X-A \mid A.\ A\in F\})$
by (*unfold c-open-def*) *auto*
— equals union of original
from *Fs-open* **and** *non-empty* **have** $(X - \bigcap\{X-A \mid A.\ A\in F\}) = \bigcup F$
by (*unfold c-open-def*) *auto*
with *open-thing* **show** *c-open* $(\bigcup F)$
by *simp*
qed

theorem (**in** *closure-space*) *four-twelve*:
topological-space X $\{A.\ c\text{-}open\ A\}$
proof (*intro topological-space.intro*)
from *non-empty* **show** $X \neq \{\}$ **.**
have *c-closed* $\{\}$

```
    by (rule four-eleven)
  thus X ∈ {A. c-open A}
    by (unfold c-open-def) auto
  have c-closed X
    by (rule four-eleven)
  thus {} ∈ {A. c-open A}
    by (unfold c-open-def) auto
  from finite-inter-open
  show ∀F. F ⊆ {A. c-open A} ∧ finite F & F≠{} ⟶ ⋂F ∈ {A. c-open A}
    by auto
  from union-open
  show ∀F. F ⊆ {A. c-open A} ⟶ ⋃F ∈ {A. c-open A}
    by auto  show ∀A∈{A. c-open A}. A ⊆ X
    by (unfold c-open-def) auto
qed

theorem (in closure-space) four-thirteen:
  assumes A-subset: A ⊆ X
  shows Cl A = ⋂{B. c-closed B & A ⊆ B}
proof
  from A-subset and c-closure-subset and idempotent
  have c-closed (Cl A)
    by (simp add: c-closed-def)
  moreover from A-subset and non-decreasing
  have A ⊆ Cl A
    by simp
  ultimately have Cl A ∈ {B. c-closed B & A ⊆ B}
    by simp
  thus ⋂{B. c-closed B ∧ A ⊆ B} ⊆ Cl A
    by auto
next
  have ∀B∈{B. c-closed B ∧ A ⊆ B}. Cl A ⊆ B
  proof (rule, simp only: mem-Collect-eq, elim conjE)
    fix B
    assume A-sub-B: A ⊆ B
    assume c-closed B
    hence  B-subset: B ⊆ X
      and B-fixed: Cl B = B
      by (simp-all!)
    from A-subset and B-subset and A-sub-B
    have Cl A ⊆ Cl B
      by (rule four-ten)
    with B-fixed show Cl A ⊆ B
      by auto
  qed
  thus Cl A ⊆ ⋂{B. c-closed B ∧ A ⊆ B}
    by auto
qed
```

3 Equivalence Theorems

Mendelson does not state the main results that we are interested in as theorems, though he does dedicate a couple of paragraphs to explaining how they follow from preceding results. We informally state the theorem, and reproduce Mendelsons discussion, before completing the formal development.

Theorem 1. *For any given set, the topological closures and the topological spaces over that set are in one to one correspondence.*

Proof. (Mendelson [1, Page 99], with my comments in braces)

Now, suppose we start with a topological space (X, τ). By Theorem 4.7 this yields a closure space. By Theorem 4.6 the closed subsets of the topological space are precisely the same as the closed subsets of the closure space, hence the same is true of open sets. It follows that the closure space we have constructed gives rise to the original topological space (X, τ).

On the other hand, had we started with a closure space and by virtue of Corollary 4.12 defined a topological spce, then by comparing Theorems 4.5 and 4.13 we see that the closure operation is the same in both spaces; that is, the topological space gives rise to the original closure space.

□

lemma *topology-to-closure-to-topology*:
  includes *topological-space*
  includes *closure-space*
  assumes Cl = *topo-closure*
  shows $\{A.$ *c-open* $A\} = T$
proof (*rule set-ext*, *simp only*: *mem-Collect-eq*, *intro iffI*)
  fix A
  assume *A-c-open*: *c-open* A
  hence $-(X - A) = X - A$ **and** $X - A \subseteq X$
    by (*simp-all*!)
  with *four-six* **have** *t-closed* $(X - A)$
    by *blast*
  from *this* **obtain** A'
    where *A′-open*: $A' \in T$
    and *same-diff*: $X - A = X - A'$
    by (*unfold t-closed-def*) *auto*
  from *A′-open* **and** *carrier-subset*
  have $A \subseteq X$ **and** $A' \subseteq X$
    by (*simp-all*!)
  with *same-diff* **have** $A = A'$

```
    by auto
  with A'-open show A∈T
    by auto
next
  fix A
  assume A ∈ T
  with carrier-subset have subset: A⊆X
    by (simp!)
  hence comp-subset: X−A ⊆ X and t-closed (X−A)
    by (simp-all!, auto)
  with four-six have _(X−A) = X−A
    by blast
  hence Cl (X−A) = X−A
    by (simp!)
  with comp-subset have c-closed (X−A)
    by (simp!)
  with subset show c-open A
    by (simp!)
qed


lemma closure-to-topology-to-closure:
  includes topological-space
  includes closure-space
  assumes c-top: T = {A. c-open A}
  shows   ∀ A. A⊆X ⟶ Cl A = Ā
proof (intro allI impI)
  have closed-is-closed: t-closed = c-closed
  proof (rule, rule)
    fix A
    assume A-closed: t-closed A
    from this obtain A'
      where A'∈T
      and A-comp: A=X−A'
      by (unfold t-closed-def) auto
    with c-top have c-open A'
      by auto
    with A-comp show c-closed A
      by (simp add: c-open-def c-closed-def)
  next
    fix A
    assume 1: c-closed A
    hence A⊆X
      and Cl A = A
      by (unfold c-closed-def) auto
    hence 2: A=X−(X−A)
      and 3: X−A⊆X
      by auto
    from 1 and 2 and 3
```

have *c-open* $(X-A)$
by (*unfold c-open-def*) *auto*
with *c-top* **have** $X-A \in T$
by *simp*
with *2* **and** *3* **show** *t-closed A*
by (*unfold t-closed-def*) *auto*
qed
fix *A*
assume *A-subset*: $A \subseteq X$
hence $\overline{A} = \bigcap\{B.$ *t-closed* B & $A \subseteq B\}$
by (*simp only*: *four-five*)
also have $\ldots = \bigcap\{B.$ *c-closed* B & $A \subseteq B\}$
by (*simp only*: *closed-is-closed*)
also from *A-subset* **and** *four-thirteen*
have $\ldots =$ *Cl A*
by *auto*
finally show *Cl A* $= \overline{A}$ **..**
qed

end

4 Discussion

4.1 To Do

- Implement $\overline{A}$ notation instead of $C\ A$ (which Mendelson uses for complement).
- Include all of Mendelson's original proofs.
- Measure the (de Bruin) "loss factor".
- Investigate each deviation from Mendelson.

4.2 Formal vs Informal

I was unable to translate one direction of Menelsons proof, or even get anything that I could formalise.

References

[1] B. Mendelson. *Introduction to Topology.* Blackie & Son, 1963,1962.

[2] L. Paulson and T. Nipkow. Isabelle. http://isabelle.in.tum.de.